
1 Overview

This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2019-06_D20wk11 related to the title of this CR.

Contact:
Shaun Harry

shaun.harry@keysight.com
2 Corrections required
2.1 Change 1

	Function name
	f_NR_RRC_PSCellChange_IntraCell_CommonSS()

	Reason for change
	Following The implementation of MCC comments for TTCN CR R5s200126 (related to test case 7.1.3.4.2), the function f NR_RRC_PSCellChange_IntraCell_CommonSS() was changed according to the following comment:

MCC160 proposed implementation:

-
Delete activation and deactivation of a_WaitForRRCReconfigurationComplete

-
Wait for the coordination message at the end  of the function

-
The same change is done in f_NR_RRC_PSCellChange_IntraCell_CommonSS

The application of this approach to f_NR_RRC_PSCellChange_IntraCell_CommonSS() means that the function will always stop and wait for the coordination message containing the RRCReconfigurationComplete from EUTRA. This gives problems because in some handover scnearios this function is used in conjunction with the function f_EUTRA_508RRC_IntraLTE_HO_IntraCell_Step6_7() at the EUTRA PTC, this is a legacy EUTRA function which will receive RRCConnectionReconfigurationComplete but does not forward the embedded NR message to NR PTC for checking. This causes processing of the NR PTC to deadlock.



	Summary of change
	Revert the changes made to this function following R5s200126, so that handling of RRCReconfigurationComplete is optional.

	TTCN module
	NR_Handover_ENDC.ttcn

	MCC160 Comment
	Accepted in principle but implemented differently. The reception of RRCReconfigurationComplete on the NR side shall not be checked in a default handler because it is clearer and because the reception of that message is part of the ENDC PScell change sequence. When it is handled in a default we don’t know if it has been transmitted by UE or not in the RCConnectionReconfiguartionComplete.
Instead it is proposed to update f_EUTRA38_IntraLTE_HO_IntraPSCell_NR_PDCP.

See proposed implementation below.


Before change:

function f_NR_RRC_PSCellChange_IntraCell_CommonSS(NR_CellId_Type                           p_CellId,

                                                     RNTI_Value                               p_RNTI_Value :=  tsc_C_RNTI_Value3,

                                                     template (value) RadioBearerConfig       p_RadioBearerConfig,

                                                     template (omit) RadioBearerConfig        p_RadioBearerConfig2 := omit,

                                                     template (value) CellGroupConfig         p_SCGConfig,

                                                     boolean                                  p_RBConfig_KeyChange := true,

                                                     template (value) NR_RachProcedureConfig_Type p_RachProcedureConfig,

                                                     template (value) NR_RadioBearerList_Type p_SS_RadioBearerToReleaseList,

                                                     template (value) NR_RadioBearerList_Type p_SS_RadioBearerList,

                                                     template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit,

                                                     template (omit) MeasConfig               p_MeasConfig := omit

                                                    ) runs on ENDC_NR_PTC

  {

    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList;

    var SubFrameTiming_Type v_Timing;

    var EUTRA_NR_Coordination_MSG v_Coordination_MSG_Value;

    //Step 1 NR Cell before T: Get PDCP count for DRBs and SRB3 (if applied)

    v_NR_PdcpCountInfoList := f_NR_GetPdcpCountInfoListIntraNR_HO(p_CellId, p_SS_RadioBearerList, p_RBConfig_KeyChange);

   //Step 2 to 5 @sic R5s190337 R5s190370 R5s200126 sic@

    v_Timing := f_NR_RRC_PSCellChange_IntraCell_Step2To5 (p_CellId, p_RNTI_Value,

                                                          p_RadioBearerConfig, p_RadioBearerConfig2, p_SCGConfig,

                                                          p_SS_RadioBearerToReleaseList, p_SS_RadioBearerList,

                                                          p_RachProcedureConfig, p_RACH_ConfigDedicated,

                                                          p_MeasConfig);

    //Step 6 NR Cell at T + 5ms: Restore the PDCP counts

    //NOTE 1:   For AM DRBs the PDCP count is maintained. For SRB3 (if applied) and UM DRBs, the PDCP count is maintained or reset depending on the RRCConnectionReconfiguration message content.

    f_NR_SS_PdcpCount_Set(p_CellId, v_NR_PdcpCountInfoList, cs_TimingInfo_NR(v_Timing));

    //Step 7 NR Cell at T + 5ms: Re-establish security.

    f_ENDC_GetKeysActivateSecurity(p_CellId, EUTRA); // @sic R5s190587 sic@

    //Step 8 E-UTRA Cell (after step 2): Receive RRCConnectionReconfigurationComplete

    //RRCReconfigurationComplete not yet received from E-UTRA - Wait for it @sic R5s200126 sic@

    EUTRA.receive(cmr_EUTRA_NR_OctetData ("Complete")) -> value v_Coordination_MSG_Value;

    f_NR_CheckReceived_RRCReconfigurationComplete(valueof(v_Coordination_MSG_Value));
}
After change:

function f_NR_RRC_PSCellChange_IntraCell_CommonSS(NR_CellId_Type                           p_CellId,

                                                     RNTI_Value                               p_RNTI_Value :=  tsc_C_RNTI_Value3,

                                                     template (value) RadioBearerConfig       p_RadioBearerConfig,

                                                     template (omit) RadioBearerConfig        p_RadioBearerConfig2 := omit,

                                                     template (value) CellGroupConfig         p_SCGConfig,

                                                     boolean                                  p_RBConfig_KeyChange := true,

                                                     template (value) NR_RachProcedureConfig_Type p_RachProcedureConfig,

                                                     template (value) NR_RadioBearerList_Type p_SS_RadioBearerToReleaseList,

                                                     template (value) NR_RadioBearerList_Type p_SS_RadioBearerList,

                                                     template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit,

                                                     template (omit) MeasConfig               p_MeasConfig := omit

                                                    ) runs on ENDC_NR_PTC

  {

    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList;

    var SubFrameTiming_Type v_Timing;

    var EUTRA_NR_Coordination_MSG v_Coordination_MSG_Value;

    var default v_WaitForRRCReconfigurationComplete;
    //Step 1 NR Cell before T: Get PDCP count for DRBs and SRB3 (if applied)

    v_NR_PdcpCountInfoList := f_NR_GetPdcpCountInfoListIntraNR_HO(p_CellId, p_SS_RadioBearerList, p_RBConfig_KeyChange);

   //Step 2 to 5 @sic R5s190337 R5s190370 R5s200126 sic@

   v_WaitForRRCReconfigurationComplete := activate(a_WaitForRRCReconfigurationComplete()); //Receive Complete message at any time
    v_Timing := f_NR_RRC_PSCellChange_IntraCell_Step2To5 (p_CellId, p_RNTI_Value,

                                                          p_RadioBearerConfig, p_RadioBearerConfig2, p_SCGConfig,

                                                          p_SS_RadioBearerToReleaseList, p_SS_RadioBearerList,

                                                          p_RachProcedureConfig, p_RACH_ConfigDedicated,

                                                          p_MeasConfig);

    //Step 6 NR Cell at T + 5ms: Restore the PDCP counts

    //NOTE 1:   For AM DRBs the PDCP count is maintained. For SRB3 (if applied) and UM DRBs, the PDCP count is maintained or reset depending on the RRCConnectionReconfiguration message content.

    f_NR_SS_PdcpCount_Set(p_CellId, v_NR_PdcpCountInfoList, cs_TimingInfo_NR(v_Timing));

    //Step 7 NR Cell at T + 5ms: Re-establish security.

    f_ENDC_GetKeysActivateSecurity(p_CellId, EUTRA); // @sic R5s190587 sic@

    //Step 8 E-UTRA Cell (after step 2): Receive RRCConnectionReconfigurationComplete

    deactivate(v_WaitForRRCReconfigurationComplete);
  }
MCC160 proposed implementation
  function f_EUTRA38_IntraLTE_HO_IntraPSCell_NR_PDCP (EUTRA_CellId_Type p_CellId,

                                                      ENDC_SRB_DRB_Configuration_Type p_SecondaryConfig,

                                                      template (omit) integer p_SK_Counter := 0

                                                      ) runs on EUTRA_5GS_PTC

  { //@sic R5s190476 change 4.15: added p_SK_Counter sic@

    var template (value) RadioBearerList_Type v_DrbConfigAtSS := cs_DrbConfigList_DRB1; //@sic R5s191169 R5s200051 change 1 sic@

    var integer v_DRB_Id := f_EUTRA38_ENDC_GetDRBIdOfMCG();

    var template (value) octetstring v_NR_SCGConfig;

    var template (value) octetstring v_NR_RBConfig;

    var template (value) octetstring v_NR_RBConfig2;

    var octetstring v_OctetData;

    var template (value) RRCConnectionReconfiguration_v890_IEs v_RRCConnectionReconfiguration_v890_IEs;

    var SubFrameTiming_Type v_Timing;

    var template (value) RadioBearerList_Type v_SrbDrbList;   //@sic R5s190476 change 4.2 sic@

    var boolean v_AddRelDrb8 := false; //To include RadioResourceConfigDedicated-DRB-Rel-Add  with DRB8 in the msg sent to UE @sic R5s200051 change 1 sic@

    select (p_SecondaryConfig) {

        case (NR_PDCP_MCG_SRB1_2) {// LTE config to configure NR PDCP on EUTRA PTC

            NR.send (cms_EUTRA_NR_OctetData (int2oct(v_DRB_Id, 1), "DRB ID")); // if changing MCG DRB need to tell NR the Id

        }

    }

    select (p_SecondaryConfig) { // if changing SRB 1/2 need to also update the ciphering/integrity algorithms to NR values

      case (NR_PDCP_SRB1_2, NR_PDCP_MCG_SRB1_2, NR_PDCP_SRB1_2_SRB3) {

        v_OctetData := f_EUTRA_NR_WaitForCoOrd_Data(NR, "NR Ciphering Alg");

        f_EUTRA_AS_CipheringAlgorithm_Set(f_EUTRA_ConvertCiphAlg2Bitstring(substr(oct2bit(v_OctetData), 5, 3)));

        v_OctetData := f_EUTRA_NR_WaitForCoOrd_Data(NR, "NR Integrity Alg");

        f_EUTRA_AS_IntegrityAlgorithm_Set(f_EUTRA_ConvertIntAlg2Bitstring(substr(oct2bit(v_OctetData), 5, 3)));

      }

    }

    //Receive SCG, RBConfig and RBConfig2 from NR PTC

    v_NR_SCGConfig := f_EUTRA_NR_WaitForCoOrd_Data(NR, "Reconfig");

    v_NR_RBConfig := f_EUTRA_NR_WaitForCoOrd_Data(NR, "RB");

    v_NR_RBConfig2 := f_EUTRA_NR_WaitForCoOrd_Data(NR, "RB2");

    v_SrbDrbList := {    //@sic R5s190476 change 4.2 sic@

        cds_SRB1_Config_NRPDCP,

        cds_SRB2_Config_NRPDCP

        };

    select (p_SecondaryConfig) {

        case (NR_PDCP_SRB1_2, NR_PDCP_SRB1_2_SRB3) {// LTE config to configure NR PDCP on EUTRA PTC

            //@sic R5s190476 change 4.16 sic@

            // do nothing as SRBs changed elsewhere

          }

        case (NR_PDCP_MCG_SRB1_2) {// LTE config to configure NR PDCP on EUTRA PTC

            v_DrbConfigAtSS := { cs_RadioBearer_Reconfig_DRBwithNRPDCPonEUTRA(tsc_DRB8, cs_RLC_Configuration_AM, cs_MAC_Configuration_AM) }; //@sic R5s190554 change 4.22 sic@

            v_AddRelDrb8 := true;  //@sic R5s200051 change 1 sic@

        }

    }

    v_RRCConnectionReconfiguration_v890_IEs := f_EUTRA38_508_BuildRRCConnReconfig_NR_Config_Setup (v_NR_SCGConfig,

                                                                                                   v_NR_RBConfig,

                                                                                                   p_SK_Counter,    //@sic R5s190476 change 4.15 sic@

                                                                                                   fl_NR_GetP_Max(),

                                                                                                   v_NR_RBConfig2);

    v_Timing := f_EUTRA_GetNextSendOccasion(p_CellId, tsc_EUTRA_DelayBeforeIntraCellHO);

    f_EUTRA38_SendTimeToNR(p_CellId, v_Timing);

    f_EUTRA38_IntraLTE_HO_IntraCell_PSCellchange ( p_CellId,

                                          cs_TimingInfo_EUTRA_NB (v_Timing),

                                          v_DrbConfigAtSS,

                                          -, // tsc_C_RNTI_Def2,

                                          -, // p_MeasConfig

                                          -, // p_Dsr_TransMax  := n4,

                                          v_RRCConnectionReconfiguration_v890_IEs,

                                          v_SrbDrbList,               //@sic R5s190476 change 4.2 sic@

                                          true,                       //@sic R5s190554 change 4.5 sic@

                                          v_AddRelDrb8);              //@sic R5s190918 sic@  to include RadioResourceConfigDedicated-DRB-Rel-Add  with DRB8 in the msg sent

    f_EUTRA38_ENDC_InitiateAndSendS_gNBKeys(NR); //@sic R5s200046 change 4 sic@

  }
2.2 Change 2

	Function name
	New altstep a_WaitForRRCReconfigurationComplete()

	Reason for change
	Following The implementation of MCC comments for TTCN CR R5s200126 (related to test case 7.1.3.4.2), the function f NR_RRC_PSCellChange_IntraCell_CommonSS() was changed according to the following comment:

MCC160 proposed implementation:

-
Delete activation and deactivation of a_WaitForRRCReconfigurationComplete

-
Wait for the coordination message at the end  of the function

-
The same change is done in f_NR_RRC_PSCellChange_IntraCell_CommonSS

The application of this approach to f_NR_RRC_PSCellChange_IntraCell_CommonSS() means that the function will always stop and wait for the coordination message containing the RRCReconfigurationComplete from EUTRA. This gives problems because this function is used in conjunction with the function f_EUTRA_508RRC_IntraLTE_HO_IntraCell_Step6_7() at the EUTRA PTC, this is a legacy EUTRA function which will receive RRCConnectionReconfigurationComplete but does not forward the embedded NR message to NR PTC for checking. This causes processing of the NR PTC to deadlock.



	Summary of change
	Reinstate altsetp used prior to IWD 20wk11

	TTCN module
	NR_Handover_ENDC.ttcn

	MCC160 Comment
	Not needed – see change 1


New altstep:

altstep a_WaitForRRCReconfigurationComplete() runs on NR_BASE_PTC

  {

    [] EUTRA.receive(cmr_EUTRA_NR_OctetData ("Complete"))  {}

  }

